
Code for Procedure ‘ShowContextMenu’

      // =======================================================================
      // Function ShowContextMenu:
      // =======================================================================
      // This procedure displays a context menu based upon the data structure
      // passed.
      // =======================================================================
     P ShowContextMenu...
     P                 B                   export
     D                 PI
     D  inSystemInfo...
     D                                     likeds(tLRSystemInfoDS) const
     D  ioContextMenu...
     D                                     likeds(tLRNG_contextMenu)
     D  ioSavedState...
     D                                     likeds(tLRNG_SavedStateDS)

      // =======================================================================
      // - Local Constants -
      // =======================================================================

     D FIRST_OPTION_ROW...
     D                 C                   2

      // =======================================================================
      // - Local Variables -
      // =======================================================================

     D cnt...
     D                 S                   like(tInt) inz
     D optCnt...
     D                 S                   like(tInt) inz
     D dspCnt...
     D                 S                   like(tInt) inz

      /free

       // Count non-empty options if option count not specified
       if ioContextMenu.optCnt > 0;
          optCnt = ioContextMenu.optCnt;
       else;
          for cnt = 1 to MAX_CONTEXT_MENU_OPTIONS;
             if ioContextMenu.opts(cnt).text <> *blanks;
                optCnt += 1;
             else;
                ioContextMenu.optCnt = optCnt;
                leave;
             endif;
          endfor;
       endif;

       // Start using the container
       LRNG_UsingEx(*off : ioContextMenu.container);

          LRNG_SetProperty(
             'showPopOver' : TRUE
                           );

          // Start using the popover form
          LRNG_Using('popOver');

             LRNG_SetProperties(
                'type : form' :
                'grid.col%1.width : fit' :
                'grid.row%1.height : fit'
                                );

             ApyFmt(%trim(ioContextMenu.container) +
                    '.popOver' :
                    inSystemInfo : ioSavedState);

             // Message field
             LRNG_Using('fields.txtMsg');

                LRNG_SetProperties(
                   'type : textBox' :
                   'editable : FALSE' :
                   'value : ' + %trim(ioContextMenu.msg) :
                   'layout.col : 1' :
                   'layout.row : 1' :
                   'layout.width : fit,,250' :
                   'layout.height : fit' :
                   'multiLine : TRUE'
                                   );
                ApyFmt(%trim(ioContextMenu.container) +
                       '.popOver' +
                       '.fields.txtMsg' :
                       inSystemInfo : ioSavedState);

                // Override theme values applied
                LRNG_SetProperties(
                   'multiLine : TRUE'
                                   );

             LRNG_EndUsing ();

             // Put buttons on a new form for placement control
             LRNG_Using('fields.frmBtn');

                LRNG_SetProperties(
                   'type : form' :
                   'layout.cols : -1' :
                   'layout.col : 1' :
                   'layout.row : ' + %char(FIRST_OPTION_ROW) :
                   'layout.width : fit' :
                   'layout.height : fit' :
                   'grid.hAlign : center'
                                   );

                   ApyFmt(%trim(ioContextMenu.container) +
                          '.popOver' +
                          '.fields' +
                          '.frmBtn' :
                          inSystemInfo : ioSavedState);

                // Menu Option Fields
                for cnt = 1 to optCnt;

                   if ioContextMenu.opts(cnt).enabled = TRUE;

                      LRNG_Using('fields.btnOpt' + %char(cnt));

                         LRNG_SetProperties(
                            'type : button' :
                            'text : ' + %trim(ioContextMenu.opts(cnt).text) :
                            'hAlign : center' :
                            'layout.col : 1' :
                            'layout.row : ' +
                              %char(dspCnt + 1) :
                            'layout.width : fit' :
                            'onClick : ' +
                              %trim(ioContextMenu.opts(cnt).action)
                                            );
                         ApyFmt(%trim(ioContextMenu.container) +
                                '.popOver' +
                                '.fields.frmBtn' +
                                '.fields.btnOpt' +
                                %char(cnt) :
                                inSystemInfo : ioSavedState);

                         // Return fully qualified context of each button
                         ioContextMenu.opts(cnt).context =
                              %trim(ioContextMenu.container)
                            + '.popOver'
                            + '.fields.frmBtn'
                            + '.fields.btnOpt'
                            +  %char(cnt);

                      LRNG_EndUsing ();

                      dspCnt += 1;

                   endif;
                endfor;

                // 'Cancel' button
                LRNG_Using('fields.btnCancelContext');

                   LRNG_SetProperties(
                      'type : button' :
                      'text : Cancel' :
                      'hAlign : center' :
                      'layout.col : 1' :
                      'layout.row : ' +
                        %char(dspCnt + 2) :
                      'layout.width : fit' :
                      'onClick : BTNCANCELCONTEXTONCLICK'
                                      );

                   ApyFmt(%trim(ioContextMenu.container) +
                          '.popOver' +
                          '.fields.frmBtn' +
                          '.fields.btnCancelContext' :
                          inSystemInfo : ioSavedState);

                LRNG_EndUsing ();

             LRNG_EndUsing ();

          LRNG_EndUsing ();

       LRNG_EndUsing ();

       return;

      /end-free

     P ShowContextMenu...
     P                 E


Context menu templates:
      /if defined(tDfn_ContextMenu)
     D tLRNG_contextMenu...
     D                 DS                  qualified template
     D  container...
     D                                     like(tLRNG_container)
     D                                     inz('/form')
     D  msg...
     D                                     like(tLRNG_message)
     D                                     inz('Please select option below.')
     D  optCnt...
     D                                     like(tInt) inz
     D  opts...
     D                                     likeds(tLRNG_contextMenuOption)
     D                                     dim(MAX_CONTEXT_MENU_OPTIONS)
     D                                     inz(*likeds)

     D tLRNG_contextMenuOption...
     D                 DS                  qualified template
     D  text...
     D                                     like(tLRNG_name) inz
     D  action...
     D                                     like(tLRNG_RqstAct) inz
     D  enabled...
     D                                     like(tSw) inz(TRUE)
     D  context...
     D                                     like(tLRNG_UIContext) inz
      /undefine tDfn_ContextMenu
      /endif

